

Једначине стања, импулсни одзив

Figure 1:

Вредности елемената електричног кола са слике су познате.

(а) Одредити једначине стања у скаларном и матричном облику.


Који је ред кола?

(б) Одредити импулсни одзив (Гринову функцију) за напон u ако је

$$R_1 = R_2 = R,$$

$$L = CR^2.$$

(в) Које су природне учестаности (сопствене учестаности, својствене учестаности) кола?



```
(%i1) jednacine: [ig = iL + i1,
                  iL = iC + i2,
                  u1 = uL + uC,
                  uC = u2,
                  u1 = R1 * i1,
                  u2 = R2 * i2,
                  uL = L * 'diff(iL,t),
                  iC = C * 'diff(uC,t)];

(%o1) [ig=iL+i1, iL=iC+i2, u1=uL+uC, uC=u2, u1=R1 i1, u2
      =R2 i2, uL=L(d/dt iL), iC=C(d/dt uC)]

(%i2) jednacineDiLDuC: jednacine, 'diff(iL,t)=DiL, 'diff(uC,t)=DuC;
(%o2) [ig=iL+i1, iL=iC+i2, u1=uL+uC, uC=u2, u1=R1 i1, u2
      =R2 i2, uL=DiL L, iC=C DuC]

(%i3) JednacineIzvoda: eliminate(jednacineDiLDuC,
                                  [i1, i2, iC, u1, u2, uL]);
(%o3) [uC-R1 ig+R1 iL+DiL L, uC-R2 iL+C DuC R2]

(%i4) jednacineStanja: linsolve(JednacineIzvoda, [DiL, DuC]);
(%o4) [DiL=- (uC-R1 ig+R1 iL)/L, DuC=- (uC-R2 iL)/(C R2)]
```

```
(%i5) jednacineDiff: jednacineStanja,
        DiL='diff(iL, t),
        DuC='diff(uC, t);

(%o5) 
$$\left[ \frac{d}{dt} iL = - \frac{uC - R1 i_g + R1 iL}{L}, \frac{d}{dt} uC = - \frac{uC - R2 iL}{C R2} \right]$$


(%i6) assume(R>0, C>0);
(%o6)  $[R>0, C>0]$ 

(%i7) zamene: [R1=R, R2=R, L=C·R^2];
(%o7)  $[R1=R, R2=R, L=C R^2]$ 

(%i8) jednacineDiffZamene:jednacineDiff, zamene,
        ig=unit_step(t), [iL=iL(t), uC=uC(t)];

(%o8) 
$$\left[ \frac{d}{dt} iL(t) = - \frac{-R \text{unit\_step}(t) + uC(t) + R iL(t)}{C R^2}, \right.$$


$$\left. \frac{d}{dt} uC(t) = - \frac{uC(t) - R iL(t)}{C R} \right]$$


(%i9) atvalue(iL(t), t=0,0);
(%o9) 0

(%i10) atvalue(uC(t), t=0,0);
(%o10) 0

(%i11) odziv: desolve(jednacineDiffZamene, [iL(t), uC(t)]);

(%o11) 
$$[ iL(t) = \frac{e^{-\frac{t}{C R}} \left( \frac{C^2 R^2 \sin\left(\frac{t}{C R}\right)}{2} - \frac{C^2 R^2 \cos\left(\frac{t}{C R}\right)}{2} \right)}{C^2 R^2} + \frac{1}{2},$$


$$uC(t) = \frac{e^{-\frac{t}{C R}} \left( -\frac{C^2 R^3 \sin\left(\frac{t}{C R}\right)}{2} - \frac{C^2 R^3 \cos\left(\frac{t}{C R}\right)}{2} \right)}{C^2 R^2} + \frac{R}{2} ]$$

```

ОДСКОЧНИ ОДЗИВ

```
(%i12) f:uC(t), odziv, ratsimp;

(%o12) 
$$- \frac{e^{-\frac{t}{C R}} \left( R \sin\left(\frac{t}{C R}\right) + R \cos\left(\frac{t}{C R}\right) - R e^{\frac{t}{C R}} \right)}{2}$$

```

Импулсни одзив

```
(%i13) g: diff(f,t), ratsimp;
```

```
(%o13) 
$$\frac{e^{-\frac{t}{CR}} \sin\left(\frac{t}{CR}\right)}{C}$$


```

```
(%i14) vrednosti: [R=1000, C=10^(-6)];
```

```
(%o14) [R=1000, C= $\frac{1}{1000000}$ ]
```

```
(%i15) wxplot2d([ev(g, vrednosti)], [t,0,0.01],
  [xlabel," t [s] "],
  [ylabel," g_uc "],
  [legend, "Impulsni odziv"], grid2d)$
```

```
(%t15)
```



Једначине стања у матричном облику

```
(%i16) jednacineDiffExpand: expand(jednacineDiff);
```

```
(%o16) [ $\frac{d}{dt} iL = -\frac{uC}{L} + \frac{R1 ig}{L} - \frac{R1 iL}{L}$ ,  $\frac{d}{dt} uC = \frac{iL}{C} - \frac{uC}{CR2}$ ]
```

```
(%i17) rhsJenacineDiff: map(rhs, jednacineDiffExpand);
```

```
(%o17) [ $-\frac{uC}{L} + \frac{R1 ig}{L} - \frac{R1 iL}{L}$ ,  $\frac{iL}{C} - \frac{uC}{CR2}$ ]
```

```
(%i18) Acol1: map(lambda([x], coeff(x,iL,1)), rhsJenacineDiff);
```

```
(%o18)  $\left[-\frac{R1}{L}, \frac{1}{C}\right]$ 
```

```
(%i19) Acol2: map(lambda([x], coeff(x,uC,1)), rhsJenacineDiff);
```

```
(%o19)  $\left[-\frac{1}{L}, -\frac{1}{C R2}\right]$ 
```

```
(%i20) Atr:matrix(Acol1, Acol2);
```

```
(%o20) 
$$\begin{pmatrix} -\frac{R1}{L} & \frac{1}{C} \\ -\frac{1}{L} & -\frac{1}{C R2} \end{pmatrix}$$

```

```
(%i21) A: transpose(Atr);
```

```
(%o21) 
$$\begin{pmatrix} -\frac{R1}{L} & -\frac{1}{L} \\ \frac{1}{C} & -\frac{1}{C R2} \end{pmatrix}$$

```

```
(%i22) X:matrix([iL],[uc]);
```

```
(%o22) 
$$\begin{pmatrix} iL \\ uc \end{pmatrix}$$

```

```
(%i23) Ftr: rhsJenacineDiff, iL=0, uC=0;
```

```
(%o23)  $\left[\frac{R1 \text{ ig}}{L}, 0\right]$ 
```

```
(%i24) F: transpose(matrix(Ftr));
```

```
(%o24) 
$$\begin{pmatrix} \frac{R1 \text{ ig}}{L} \\ 0 \end{pmatrix}$$

```

```
(%i25) print('diff(X,t), "=", A, "*", X, "+", F) $
```

$$\frac{d}{dt} \begin{pmatrix} iL \\ uc \end{pmatrix} = \begin{pmatrix} -\frac{R1}{L} & -\frac{1}{L} \\ \frac{1}{C} & -\frac{1}{C R2} \end{pmatrix} \cdot \begin{pmatrix} iL \\ uc \end{pmatrix} + \begin{pmatrix} \frac{R1 \text{ ig}}{L} \\ 0 \end{pmatrix}$$